
1. Rangos con Pares de Elementos de Arreglo

Ejemplo (p5, ej. 9): Dado un arreglo A : array[0, N) of Num con N ≥ 0,
contar cuántas veces coinciden dos elementos:

Const N : Int, A : array [0, N) of Int;
Var r : Int;

{P : N ≥ 0}
S
{Q : r = 〈N i, j : 0 ≤ i < j < N : A.i = A.j 〉}

Es evidente que se necesita un ciclo para recorrer el arreglo.
Usando la técnica de reemplazo de constante por variable, se crea una nueva

variable n : Int y se obtienen el invariante y la guarda:

I ≡̇ r = 〈N i, j : 0 ≤ i < j < n : A.i = A.j 〉 ∧ 0 ≤ n ≤ N

B ≡̇ n < N

De esta forma queda garantizado el requisito I ∧ ¬B ⇒ Q.
El programa será de la forma:
Const N : Int, A : array[0, N) ofNum;
Var r, n : Int;

{P : N ≥ 0}
S0 ;
{I}
do n < N →
{I ∧B} S1 {I}

od
{Q : r = 〈N i, j : 0 ≤ i < j < N : A.i = A.j 〉}
Falta derivar S0 (inicialización) y S1 (cuerpo del ciclo).

1.1. Inicialización (S0)

Probamos con S0 de la forma n, r := 0, E con E siendo una incógnita a elegir
tal que valga la terna:

{P} n, r := 0, E {I}

O, equivalentemente,
P ⇒ wp.(n, r := 0, E).I

Luego, tomamos como hipótesis P y vemos la wp:

wp.(n, r := 0, E).I
≡ { definición wp para := }

E = 〈N i, j : 0 ≤ i < j < 0 : A.i = A.j 〉 ∧ 0 ≤ 0 ≤ N
≡ { hipótesis P : N ≥ 0 }

E = 〈N i, j : 0 ≤ i < j < 0 : A.i = A.j 〉
≡ { rango vaćıo }

E = 0

≡ { elijo E = 0 }
True

1.2. Cuerpo del Ciclo (S1)

Probamos con S1 de la forma n, r := n + 1, E con E siendo una incógnita a
elegir tal que valga la terna:

{I ∧B} n, r := n + 1, E {I}

O, equivalentemente,

I ∧B ⇒ wp.(n, r := n + 1, E).I

Luego, tomamos como hipótesis I ∧B y vemos la wp:

wp.(n, r := n + 1, E).I
≡ { definición wp para := }

E = 〈N i, j : 0 ≤ i < j < n + 1 : A.i = A.j 〉 ∧ 0 ≤ n + 1 ≤ N
≡ { 0 ≤ n + 1 ≤ N vale por hipótesis 0 ≤ n y n < N }

E = 〈N i, j : 0 ≤ i < j < n + 1 : A.i = A.j 〉
≡ { partición de rango con j < n ∨ j = n. eliminación de variable con j = n. }

E = 〈N i, j : 0 ≤ i < j < n : A.i = A.j 〉+ 〈N i : 0 ≤ i < n : A.i = A.n 〉
≡ { hipótesis para r }

E = r + 〈N i : 0 ≤ i < n : A.i = A.n 〉

Acá, no hay forma de elegir una expresión E válida en el lenguaje de progra-
mación. La asignación necesita una hipótesis adicional en la precondición, que
diga que una nueva variable auxiliar s contiene el valor de la cuantificación
que sobra:

s = 〈N i : 0 ≤ i < n : A.i = A.n 〉 .

Para garantizar esta hipótesis, debe introducirse una nueva sentencia que
calcule el valor de s antes de la asignación. Luego, se replantea el cuerpo
del ciclo S1 de la siguiente forma:
{I ∧B}
S2 ;
{I ∧B ∧ s = 〈N i : 0 ≤ i < n : A.i = A.n 〉}
n, r := n + 1, E
{I}
Con la nueva hipótesis, ya se puede derivar la asignación. Tomamos como

hipótesis I ∧B ∧ s = 〈N i : 0 ≤ i < n : A.i = A.n 〉 y vemos la wp:

wp.(n, r := n + 1, E).I
≡ { mismos pasos que antes }

E = r + 〈N i : 0 ≤ i < n : A.i = A.n 〉
≡ { nueva hipótesis para s }

E = r + s

≡ { elijo E = r + s }
True

1.2.1. Ciclo Anidado (S2)

Sólo queda derivar S2. Llamamos P ′ a la precondición y Q′ a la postcondición
de S2:
{P ′ : I ∧B}
S2 ;
{Q′ : I ∧B ∧ s = 〈N i : 0 ≤ i < n : A.i = A.n 〉}

Acá, S2 debe preservar la validez de I ∧B, y por lo tanto no debe modificar
las variables r y n. Es decir, r y n son constantes para S2.

Para calcular el valor de s, es evidente que se necesita un ciclo que
recorra los primeros n elementos del arreglo. Luego se puede usar la
técnica de reemplazo de constante por variable, creando una nueva variable
m : Int y obteniendo invariante y guarda:

I ′ ≡̇ I ∧B ∧ s = 〈N i : 0 ≤ i < m : A.i = A.n 〉 ∧ 0 ≤ m ≤ n

B′ ≡̇ m < n

Observación importante: Sólo se reemplaza la ocurrencia de n en el rango de
la nueva cuantificación. No se reemplaza en el término ni en ningún otra parte.

Luego, se derivan inicialización y cuerpo de este ciclo para obtener (queda
como ejercicio):

m, s := 0, 0 ;
do m < n →

if A.m = A.n →
m, s := m + 1, s + 1
A.m 6= A.n →
m := m + 1

fi
od ;

1.3. Resultado Final

Const N : Int, A : array[0, N) ofNum;
Var r, n,m, s : Int;

{P : N ≥ 0}
n, r := 0, 0 ;
do n < N →

m, s := 0, 0 ;
do m < n →

if A.m = A.n →
m, s := m + 1, s + 1
A.m 6= A.n →
m := m + 1

fi
od ;
n, r := n + 1, r + s ;

od
{Q : r = 〈N i, j : 0 ≤ i < j < N : A.i = A.j 〉}

